

# Ecuaciones Diferenciales Ordinarias (EDOs)

## Resolución simbólica y algunas interpretaciones gráficas

*Realizado por el Prof. Pedro González Rodelas  
Dpto. de Matemática Aplicada. Univ. de Granada*

Fecha de la última revisión: 14/04/2015

Cualquier notificación de errata o sugerencia puede ser enviada al email: [prodelas@ugr.es](mailto:prodelas@ugr.es)

### ■ Paquetes externos necesarios (usando la versión 3.0 de Mathematica)

```
In[1]:= << Graphics`ImplicitPlot`
```

```
In[2]:= << Graphics`PlotField`
```

### ■ Resolución simbólica de EDOs

*En esta sección nos centraremos en la posible resolución analítica, aprovechando las capacidades de cálculo simbólico que proporciona Mathematica, de algunos tipos de Ecuaciones Diferenciales Ordinarias (EDOs), tanto lineales como no lineales.*

*Pero hay que dejar muy claro en este punto que, salvo las EDOs lineales y muy pocos tipos de las No Lineales, la mayoría de las Ecuaciones Diferenciales son bastante difíciles de resolver analíticamente, a no ser que se consiga identificarlas y clasificar como de alguno de los tipos que ya se tiene bien estudiados y se sabe como resolver, mediante técnicas específicas y algo elaboradas, en alguno de los casos. En todos los demás casos, casi siempre nos tendremos que conformar con realizar un pequeño estudio teórico previo relacionado con la posible existencia y unicidad de solución del problema en cuestión, seguido de una resolución numérica para calcular al menos una aproximación lo suficientemente buena del problema.*

### ■ Mediante el comando DSolve

*Mathematica dispone del comando **DSolve** para intentar la resolución simbólica de la Ecuación Diferencial que le indiquemos, ya sea sin o con condiciones iniciales o de contorno añadidas. Veámos a continuación cuál sería su sintaxis genérica.*

In[3]:= ? DSolve

DSolve[eqn, y, x] solves a differential equation for the function y, with independent variable x. DSolve[{eqn1, eqn2, ... }, {y1, y2, ... }, x] solves a list of differential equations. DSolve[eqn, y, {x1, x2, ... }] solves a partial differential equation.

Como vemos en la descripción general que nos indica la ayuda de Mathematica, este comando necesitaría tres tipos de argumentos bien diferenciados para su buen funcionamiento:

- en "eqn" deberíamos indicar tanto la(s) ecuación(es) diferencial(es) que queramos resolver como la(s) condición inicial(es) y de contorno que hubiera en nuestro problema, todo ello entre llaves y separado por comas;

- en "y" tendremos que indicar cuál es realmente nuestra(s) función(es) incógnita, dependiente(s) de la variable(s) independiente(s) correspondiente(s), con la letra y formato que aparezca en la(s) ecuación(es) diferencial(es) del argumento anterior; en caso de que sean varias habría que incluirlas entre llaves y separadas por comas, como suele ser habitual en Mathematica.

- en "x" indicaríamos simplemente la(s) letra(s) que hayamos usado para la(s) variable(s) independiente(s) de nuestro problema, ya sean una o varias.

Veámos algunos ejemplos, tanto de Ecuaciones diferenciales genéricas, como de Problemas de Valores Iniciales (PVI) o de Contorno (PCs):

## ■ Ecuaciones Diferenciales en general

Resolvamos por ejemplo la EDO  $y' - y = e^x$ ; nótese que dentro de la sentencia **DSolve** de hay que especificar la dependencia de la variable dependiente de la independiente y que la ecuación diferencial hay que indicarla con doble igual, como cualquier otra ecuación en Mathematica. Nótese también que aunque hayamos escrito en el texto la función exponencial con la letra  $e^x$ , dentro de la orden de Mathematica no se nos puede olvidar usar la letra especial que hay dentro de la paleta de símbolos, o bien cambiar la letra "e" minúscula por la mayúscula, que ya si sería correctamente interpretada.

In[4]:= DSolve[y'[x] - y[x] == E^x, y[x], x]

Out[4]= {{y[x] -> E^x x + E^x C[1]}}

obteniendo la solución genérica  $y(x) = x e^x + c_1 e^x$ . Nótese que Mathematica irá denotando  $C[1]$ ,  $C[2]$ , ... las distintas constantes de las que dependa la solución, dependiendo casi siempre del orden de la correspondiente ecuación diferencial de la que procede.

Veámos ahora otro ejemplo, con una ecuación diferencial de segundo orden  $y'' = y + y' + \sin x$ , en la que la solución general va a depender de dos constantes en vez de sólo una

In[5]:= DSolve[y''[x] == y[x] + y'[x] + Sin[x], y[x], x] // Simplify

Out[5]= {{y[x] ->  $\frac{1}{5} (5 E^{\frac{1}{2}(x-\sqrt{5} x)} (C[1] + E^{\sqrt{5} x} C[2]) + \cos[x] - 2 \sin[x])$ }}

Conviene realizar una cierta simplificación (a través de la orden de Mathematica **Simplify**, que podemos aplicar fácilmente tras la orden **DSolve** con la doble barra inclinada //).

Veámos ahora otro ejemplo de EDO no lineal (nótese la expresión  $x(t)^3$  en uno de los términos de la ec. dif.)  $x'(t) + t x(t) = t^3 x(t)^3$ , en la que la solución general tiene varias ramas, además de depender de una constante  $c_1$ ,

$$x(t) = \pm \frac{1}{\sqrt{1+t^2 - 2 c_1 e^{t^2}}}$$

```
In[6]:= DSolve[x'[t] + t x[t] == t^3 x[t]^3, x[t], t]
```

```
Out[6]= {{x[t] -> - 1 / Sqrt[1 + t^2 - 2 E^t^2 C[1]]}, {x[t] -> 1 / Sqrt[1 + t^2 - 2 E^t^2 C[1]]}}
```

## ■ Problemas de Valores Iniciales (PVI)

En estos casos se usaría el comando **DSolve** con el siguiente formato (asumiendo que la variable dependiente es "y" y la independiente "x", aunque otras muchas posibilidades se podrían dar)

```
DSolve[{ecuación diferencial, condición inicial}, y[x], x]
```


Resolvamos por ejemplo el PVI que se obtiene al añadir la siguiente condición inicial  $y(0) = 1$  a la ecuación diferencial resuelta anteriormente.

```
In[7]:= sol = DSolve[{y'[x] - y[x] == e^x, y[0] == 1}, y[x], x]
```

```
Out[7]= {{y[x] -> E^x (1 + x)}}
```

Podemos ahora dibujar fácilmente la solución obtenida mediante el comando **Plot**, indicándole un cierto intervalo de la variable independiente. Nótese que hemos aprovechado el comando de asignación que posee Mathematica "/" y que permite asignarle a la función "y[x]" de la parte izquierda la solución obtenida anteriormente y que habíamos denominado "sol". El hecho de indicar "sol[[1]]" es porque hemos escogido la primera y única solución; si hubiera habido dos o más soluciones, éstas vendrían numeradas "sol[[2]]", "sol[[3]]", ...

```
In[8]:= Plot[y[x] /. sol[[1]], {x, 0, 2}]
```



```
Out[8]= -Graphics-
```

Nótese que lo que hemos hecho es simplemente añadir dicha condición inicial en forma de nueva ecuación (con doble igual y usando corchetes en vez de paréntesis, ya que en Mathematica así se indica cualquier argumento de una función) dentro de la orden **DSolve**. Así hemos obtenido ya una solución concreta y única  $y(x) = e^x (1 + x)$ , que ya no depende de ninguna constante. En realidad es cómo si hubiésemos calculado el valor que debe tomar la constante  $c_1$  en el caso anterior para que se cumpla la condición inicial impuesta; es decir


$$1 = y(0) = 0 e^0 + c_1 e^0 = c_1.$$

Resolvamos ahora un PVI asociado a la EDO no lineal vista anteriormente, donde al imponer la condición inicial concreta ya se obtiene una única solución y se descarta la otra rama de posibles soluciones. Para realizar la gráfica de la correspondiente solución usamos la misma orden de asignación que antes "/" pero si no nombramos a la solución podemos simplemente usar el resultado justo anterior, con el símbolo del porcentaje "%".

```
In[9]:= DSolve[{x'[t] + t x[t] == t^3 x[t]^3, x[0] == 1/2}, x[t], t]
```

```
Plot[x[t] /. %[[1]], {t, 0, 1}]
```

```
Out[9]= {{x[t] -> 1 / Sqrt[1 + 3 E^2 + t^2]}}
```



```
Out[10]= -Graphics-
```

## ■ Problemas de Contorno (PCs)

La sintaxis general sería en este caso (asumiendo "y" y "x" como variables dependiente e independiente, respectivamente)

```
DSolve[{ecuación diferencial, condiciones de contorno}, y[x], x]
```

Veámos ahora un ejemplo de resolución de un problema de contorno, que al tener dos condiciones adicionales, tiene que venir a partir de una ecuación diferencial de segundo orden.

$$y'' = y + y' + \sin x, \quad x \in [0, \pi],$$


$$y(0) = 0, \quad y(\pi) = 1.$$

```
In[11]:= sol = DSolve[{y''[x] == y[x] + y'[x] + Sin[x], y[0] == 0, y[π] == 1}, y[x], x] // Simplify
```

```
Out[11]= {{y[x] -> \frac{1}{5(-1 + E^{\sqrt{5} \pi})} (-6 E^{\frac{1}{2}(-1 + \sqrt{5})(\pi - x)} + E^{\frac{1}{2}(1 + \sqrt{5})x} - E^{\frac{1}{2}(2\sqrt{5} \pi + x - \sqrt{5} x)} + 6 E^{\frac{1}{2}((-1 + \sqrt{5})\pi + (1 + \sqrt{5})x)} + (-1 + E^{\sqrt{5} \pi}) Cos[x] - 2(-1 + E^{\sqrt{5} \pi}) Sin[x])}}
```

Podemos ahora realizar la correspondiente gráfica, indicando con la opción "PlotRange->All" que se dibuje todo el rango posible de la variable dependiente en el intervalo indicado de la variable independiente.

```
In[12]:= Plot[y[x] /. sol[[1]], {x, 0, π}, PlotRange -> All]
```



```
Out[12]= -Graphics-
```

## ■ EDOs de primer orden

### ■ Ecuación lineal de primer orden

Una ecuación diferencial ordinaria se dice lineal de primer orden, si se puede escribir de la siguiente forma

$$y'(x) + a(x)y(x) = b(x)$$

con  $a(x)$  y  $b(x)$  sendas funciones continuas, definidas en cierto intervalo  $I$  de la recta real.

En el caso particular de que se anule  $b(x) = 0$ ,  $\forall x \in I \subset \mathbb{R}$ , se dice que dicha EDO es homogénea y se puede resolver escribiendo  $y'(x)/y(x) = -a(x)$  e integrando ambos miembros, respecto a la variable independiente  $x$ , obteniéndose  $\ln(y(x)) = -A(x)$ , siendo  $A'(x) = a(x)$  una primitiva. Por tanto, cualquier función  $y \equiv y(x) = Ce^{-A(x)}$  (con  $C \in \mathbb{R}$ , cte.) será una solución genérica de la ecuación homogénea. Y bastará con imponer una cierta condición adicional, del tipo  $y(x_0) = y_0$  por ejemplo, para determinar de forma unívoca una determinada solución, que satisfará el correspondiente PVI.

Resolvamos por ejemplo el siguiente PVI lineal homogéneo:  $y' = -k y$ ,  $y(0) = 1$ , que puede provenir de un cierto problema de transmisión de calor (según la ley de calentamiento/enfriamiento de Newton), un modelo Malthusiano de crecimiento de poblaciones, o bien de un típico problema de desintegración radioactiva.

```
In[13]:= DSolve[{y'[x] == -k y[x], y[0] == 1}, y[x], x]
```

```
Out[13]= {{y[x] -> E^{-k x}}}
```

En el caso no homogéneo, también resulta fácil comprobar que la solución de la EDO lineal de primer orden se puede expresar de la siguiente forma

$$y \equiv y(x) = (c + B(x))e^{-A(x)} \text{ (con } c \in \mathbb{R}, \text{ cte.)}$$

siendo  $A(x)$  cualquier primitiva de  $a(x)$  y  $B(x)$  una primitiva de  $b(x)e^{A(x)}$ .

```
In[14]:=
```

```
Clear[x, y, c, B]
y[x_] := (c + B[x]) e^{-A[x]}
```

```
In[16]:= y'[x] + a[x] y[x] - b[x] (* comprobación *)
```

```
% /. {A'[x] -> a[x]}
```

```
% /. {E^{-A[x]} B'[x] -> b[x]}
```

```
Out[16]= -b[x] + E^{-A[x]} a[x] (c + B[x]) - E^{-A[x]} (c + B[x]) A'[x] + E^{-A[x]} B'[x]
```

```
Out[17]= -b[x] + E^{-A[x]} B'[x]
```

```
Out[18]= 0
```

Resolvamos por ejemplo el siguiente PVI lineal no homogéneo:  $y' + 2x y = x$ ,  $y(0) = 1$ .

```
In[19]:= Clear[x, a, b, A, B]
```

```
a[x_] := 2 x
```

```
b[x_] := x
```

```
In[22]:= A[x] = \int a[x] dx
```

```
B[x] = \int b[x] e^{A[x]} dx
```

```
Out[22]= x^2
```

```
Out[23]= \frac{E^{x^2}}{2}
```

```
In[24]:= y[x] = (c + B[x]) e^{-A[x]}
```

```
y[x]
```

```
Out[24]= E^{-x^2} \left( c + \frac{E^{x^2}}{2} \right)
```

```
Out[25]= E^{-x^2} \left( c + \frac{E^{x^2}}{2} \right)
```

```
In[26]:= x0 = 0; y0 = 1;
          y[x] /. x -> x0
          Solve[% == y0, c]
          y[x] /. %[[1]]
```

```
Out[26]=  $\frac{1}{2} + c$ 
```

```
Out[27]=  $\left\{ \left\{ c \rightarrow \frac{1}{2} \right\} \right\}$ 
```

```
Out[28]=  $E^{-x^2} \left( \frac{1}{2} + \frac{E^{x^2}}{2} \right)$ 
```

*Y comprobemos también el resultado mediante la orden **DSolve***

```
In[29]:= Clear[x, y]
          DSolve[{y'[x] + a[x] y[x] == b[x], y[x0] == y0}, y[x], x]
```

```
Out[30]=  $\left\{ \left\{ y[x] \rightarrow \frac{1}{2} E^{-x^2} (1 + E^{x^2}) \right\} \right\}$ 
```

## ■ Ecuaciones no lineales de orden uno (caso no trivial)

*Cuando se trata de resolver EDOs no lineales, ya la cuestión no resulta tan fácil, tanto para nosotros como para Mathematica. Intentemos resolver por ejemplo el siguiente PVI para la EDO no lineal de primer orden*

$$y'(x) = - \left( \frac{y(x)}{x} \right)^2, y(0) = 1$$

*que podemos empezar a transformar de la siguiente forma  $\frac{y'(x)}{y(x)^2} = -\frac{1}{x^2}$ , que integrando quedaría  $\frac{1}{y(x)} = \frac{-1}{x} + c$ , o bien  $y(x) = \frac{x}{c x - 1}$ .*

*Pero esta familia de soluciones no cumpliría la condición inicial especificada, sino  $y(0)=0$ . De todas formas, ninguno de estos problemas se podría resolver sin complicaciones mediante la orden **DSolve**. Veámoslo a continuación.*

```
In[31]:= DSolve[{y'[x] == - $\left(\frac{y[x]}{x}\right)^2$ , y[0] == 1}, y[x], x]
```

```
Power::infy : Infinite expression  $\frac{1}{0}$  encountered.
```

```
Out[31]= {}
```

```
In[32]:=
```

```
Clear[x, y]
```

```
solgen = DSolve[{y'[x] == - $\left(\frac{y[x]}{x}\right)^2$ , y[0] == 0}, y[x], x]
```

```
Power::infy : Infinite expression  $\frac{1}{0}$  encountered.
```

```
Out[33]= {}
```

*Si ahora intentamos resolver la ecuación diferencial sin condición inicial alguna, Mathematica nos proporciona una solución genérica como la obtenida anteriormente de forma manual, que tampoco podríamos afirmar que es la única posible.*

```
In[34]:=
Clear[x, y]
solgen = DSolve[y'[x] == - $\left(\frac{y[x]}{x}\right)^2$ , y[x], x]
```

```
Out[35]= {{y[x] ->  $\frac{x}{-1 + x C[1]}$ }}
```


```
In[36]:= y[x] /. solgen[[1]]
% /. {x -> 0}
```

```
Out[36]=  $\frac{x}{-1 + x C[1]}$ 
```

```
Out[37]= 0
```

Y vemos que podemos encontrar por ejemplo infinitas soluciones del PVI con la condición inicial  $y(0) = 0$ .

```
In[38]:= y[x_] :=  $\frac{x}{-1 + x C[1]}$ 
Plot[{y[x] /. C[1] -> 0, y[x] /. C[1] -> -1, y[x] /. C[1] -> -2, y[x] /. C[1] -> -3}, {x, 0, 1}]
```



```
Out[39]= -Graphics-
```

Respecto a la otra condición inicial que buscábamos,  $y(0) = 1$ , no hemos podido concluir nada hasta el momento. Vemos pues claramente las dificultades con las que nos podemos encontrar al tratar con ecuaciones diferenciales de tipo no lineal, necesitando en todo caso realizar un estudio previo acerca de la existencia y unicidad de solución, antes de poder afirmar nada con rotundidad.

## ■ Caso de variables separadas o separables

Consideremos la EDO  $(y^2 + x y^2)y' + x^2 - y x^2 = 0$ , que de forma equivalente se puede expresar también  $y^2(1+x)y' = x^2(y-1)$  o incluso  $\frac{y^2}{y-1} y' = \frac{x^2}{1+x}$ .

Este ejemplo nos muestra la idea subyacente detrás del tipo de ecuaciones diferenciales de variables separadas, ya que de alguna manera hemos conseguido expresar dicha ecuación mediante una igualdad en la que en cada uno de los miembros sólo aparece una de las variables. Para resolver ahora dicha ecuación diferencial, procederemos a integrar ambos miembros, cada uno respecto a la variable involucrada.



$$\text{In[40]:= expy} = \int \frac{y^2}{y-1} dy // \text{Simplify}$$

$$\text{Out[40]= } \frac{1}{2} y (2+y) + \text{Log}[-1+y]$$

$$\text{In[41]:= exprx} = \int \frac{x^2}{1+x} dx // \text{Simplify}$$

General::spell1 : Possible spelling error: new symbol name "exprx" is similar to existing symbol "expy".

$$\text{Out[41]= } \frac{1}{2} (-2+x) x + \text{Log}[1+x]$$

$$\text{In[42]:= Expand[exprx - expy] == Cte}$$

$$\text{Out[42]= } -x + \frac{x^2}{2} - y - \frac{y^2}{2} + \text{Log}[1+x] - \text{Log}[-1+y] == \text{Cte}$$

### ■ Caso de ecuaciones homogéneas

Consideremos ahora la EDO  $4x-3y+(2y-3x)y' = 0$ , que tiene la particularidad de poder ser expresada como  $y' = \frac{3y-4x}{2y-3x} = \frac{3\frac{y}{x}-4}{2\frac{y}{x}-3}$  y por ello puede considerarse como homogénea.

Bastará pues realizar el cambio de variable  $u \equiv \frac{y}{x}$ , de manera que  $y' = u'x + u = \frac{3u-4}{2u-3}$  y podemos expresar la correspondiente ecuación diferencial para la nueva variable dependiente  $u'$   $\equiv \frac{du}{dx} = \frac{1}{x} \left( \frac{3u-4}{2u-3} - u \right) = \frac{1}{x} \left( \frac{-2u^2+6u-4}{2u-3} \right)$ , que se puede resolver separando variables e integrando como antes

$$\int \frac{1}{x} dx = \int \frac{2u-3}{-2u^2+6u-4} du + \text{Cte}$$

$$\text{In[43]:= } \int \frac{1}{x} dx - \int \frac{2u-3}{-2u^2+6u-4} du == \text{Cte}$$

$$\text{Out[43]= } \frac{1}{2} \text{Log}[2-3u+u^2] + \text{Log}[x] == \text{Cte}$$

Equivalentemente,

$$2\text{Log}(x) + \text{Log}(2-3u+u^2) - 2\text{Cte} = \text{Log}(K x^2(2-3u+u^2)) = 0$$


$$\text{o bien } 1 = K x^2(2-3u+u^2) = K x^2 \left( 2 - 3\frac{y}{x} + \left(\frac{y}{x}\right)^2 \right) = K (2x^2 - 3xy + y^2)$$

$$\text{In[44]:= } y_+[x_] := \frac{3x}{2} + \frac{\sqrt{9x^2 - 4\left(2x^2 - \frac{1}{K}\right)}}{2}$$

$$y_-[x_] := \frac{3x}{2} - \frac{\sqrt{9x^2 - 4\left(2x^2 - \frac{1}{K}\right)}}{2}$$

K = 1;

Plot[{y\_+[x], y\_-[x]}, {x, -10, 10}]



Out[46]= -Graphics-

*Realicemos a continuación algunas comprobaciones*

In[47]:= K (2 x<sup>2</sup> - 3 x y\_+[x] + y\_+[x]<sup>2</sup>) // Simplify

Out[47]= 1

In[48]:=  $\partial_x y_+[x] - \frac{3 y_+[x] - 4 x}{2 y_+[x] - 3 x}$   
% // Simplify

Out[48]=  $\frac{3}{2} + \frac{x}{2\sqrt{9x^2 - 4(-1 + 2x^2)}} - \frac{-4x + 3\left(\frac{3x}{2} + \frac{1}{2}\sqrt{9x^2 - 4(-1 + 2x^2)}\right)}{-3x + 2\left(\frac{3x}{2} + \frac{1}{2}\sqrt{9x^2 - 4(-1 + 2x^2)}\right)}$

Out[49]= 0

In[50]:=  $\partial_x y_-[x] - \frac{3 y_-[x] - 4 x}{2 y_-[x] - 3 x}$   
% // Simplify

Out[50]=  $\frac{3}{2} - \frac{x}{2\sqrt{9x^2 - 4(-1 + 2x^2)}} - \frac{-4x + 3\left(\frac{3x}{2} - \frac{1}{2}\sqrt{9x^2 - 4(-1 + 2x^2)}\right)}{-3x + 2\left(\frac{3x}{2} - \frac{1}{2}\sqrt{9x^2 - 4(-1 + 2x^2)}\right)}$

Out[51]= 0

## ■ Caso de ecuaciones exactas

Consideremos la EDO  $(2xy - x^2) dx + (x^2 - 2y) dy = 0$ . Empezaremos comprobando si efectivamente se trata de una ecuación diferencial exacta

```
In[52]:= m[x_, y_] := 2 x y - x^2
          n[x_, y_] := x^2 - 2 y
          ∂y m[x, y] == ∂x n[x, y]
```

```
Out[54]= True
```

Para obtener ahora la solución, procederemos de la manera habitual cuando se trata de este tipo de ecuaciones diferenciales

```
In[55]:= f[x, y] = ∫ m[x, y] dx + h[y]
```

```
Out[55]= - $\frac{x^3}{3}$  + x2 y + h[y]
```

```
In[56]:= Solve[∂y f[x, y] == n[x, y], h'[y]]
          h[y] = ∫ (h'[y] /. %[[1]]) dy
```

```
Out[56]= {{h'[y] → -2 y}}
```

```
Out[57]= -y2
```

```
In[58]:= f[x, y]
```

```
Out[58]= - $\frac{x^3}{3}$  + x2 y - y2
```

Así pues, la solución implícita de la ecuación diferencial sería

$$-\frac{x^3}{3} + x^2 y - y^2 = K \quad (\text{con } K \in \mathbb{R}, \text{cte.})$$

y, podríamos definir como en algún caso anterior

```
In[59]:= y+[x_] :=  $\frac{x^2}{2} + \frac{\sqrt{x^4 - 4(\frac{x^3}{3} + K)}}{2}$ 
```


```
          y-[x_] :=  $\frac{x^2}{2} - \frac{\sqrt{x^4 - 4(\frac{x^3}{3} + K)}}{2}$ 
```

```
In[61]:= NRoots[x4 - 4( $\frac{x^3}{3}$  + K) == 0, x]
```


```
Out[61]= x == -1.16921 || x == 0.296956 - 1.30547 I || x == 0.296956 + 1.30547 I || x == 1.90863
```

```
In[62]:= x1 = -1.16921063795860136`
          x2 = 1.90863201469387568`;
```

```
In[63]:= K = 1;  
graf1 = Plot[{y+[x], y-[x]}, {x, -10, x1}]  
graf2 = Plot[{y+[x], y-[x]}, {x, x2, 10}]
```




Out[63]= -Graphics-



Out[64]= -Graphics-

```
In[65]:= Show[graf1, graf2]
```



Out[65]= -Graphics-

Realicemos a continuación algunas comprobaciones

In[66]:=

$$\left(-\frac{x^3}{3} + x^2 y_+[x] - y_+[x]^2 - K\right) // \text{Simplify}$$

$$\left(-\frac{x^3}{3} + x^2 y_-[x] - y_-[x]^2 - K\right) // \text{Simplify}$$

Out[66]= 0

Out[67]= 0

In[68]:=  $\partial_x y_+[x] - \frac{2x y_+[x] - x^2}{2 y_+[x] - x^2}$   
% // Simplify

$$\text{Out[68]}= x + \frac{-4x^2 + 4x^3}{4\sqrt{x^4 - 4\left(1 + \frac{x^3}{3}\right)}} - \frac{-x^2 + 2x\left(\frac{x^2}{2} + \frac{1}{2}\sqrt{x^4 - 4\left(1 + \frac{x^3}{3}\right)}\right)}{-x^2 + 2\left(\frac{x^2}{2} + \frac{1}{2}\sqrt{x^4 - 4\left(1 + \frac{x^3}{3}\right)}\right)}$$

Out[69]= 0

In[70]:=  $\partial_x y_-[x] - \frac{x^2 - 2x y_-[x]}{x^2 - 2y_-[x]}$   
% // Simplify

$$\text{Out[70]}= x - \frac{-4x^2 + 4x^3}{4\sqrt{x^4 - 4\left(1 + \frac{x^3}{3}\right)}} - \frac{x^2 - 2x\left(\frac{x^2}{2} - \frac{1}{2}\sqrt{x^4 - 4\left(1 + \frac{x^3}{3}\right)}\right)}{x^2 - 2\left(\frac{x^2}{2} - \frac{1}{2}\sqrt{x^4 - 4\left(1 + \frac{x^3}{3}\right)}\right)}$$

Out[71]= 0

## ■ Uso de factores integrantes

Consideremos la EDO  $(y - 8x y^2)dx + (3x - 16x^2 y)dy = 0$ , que podría ser o no exacta. Lo comprobaremos y en caso negativo, la técnica del factor integrante trata de proporcionarnos un factor multiplicativo que ayude a hacerla de ese tipo.

In[72]:=  $m[x_, y_] := y - 8x y^2$   
 $n[x_, y_] := 3x - 16x^2 y$   
 $\partial_y m[x, y] == \partial_x n[x, y]$

Out[74]=  $1 - 16xy == 3 - 32xy$

Vemos claramente que no lo es, pero vamos a comprobar si admitiría un factor integrante de la forma  $\mu \equiv \mu(x)$ , para lo cual la siguiente expresión debería depender tan sólo de la variable independiente  $x$ .

In[75]:=  $\text{expr} = \text{Simplify}\left[\frac{\partial_y m[x, y] - \partial_x n[x, y]}{n[x, y]}\right]$

Out[75]=  $\frac{-2 + 16xy}{x(3 - 16xy)}$

Probemos ahora con un factor integrante de la forma  $\mu \equiv \mu(y)$ , para lo cual la siguiente expresión debería depender tan sólo de la variable independiente  $y$ .

$$\text{In}[76]:= \text{expy} = \text{Simplify}\left[\frac{\partial_y m[x, y] - \partial_x n[x, y]}{m[x, y]}\right]$$

$$\text{Out}[76]= -\frac{2}{y}$$

En este caso, sí que ha habido suerte y podemos buscar el factor integrante  $\mu \equiv \mu(y)$  gracias a la fórmula siguiente

$$\text{In}[77]:= \mu[y] = \text{Exp}\left[-\int \text{expy} \, dy\right]$$

$$\text{Out}[77]= y^2$$

$$\text{In}[78]:= \mu[y]$$

$$\text{Out}[78]= y^2$$

$$\begin{aligned} \text{In}[79]:= & \text{my}[x_, y_] := m[x, y] \mu[y] \\ & \text{ny}[x_, y_] := n[x, y] \mu[y] \\ & \partial_y \text{my}[x, y] - \partial_x \text{ny}[x, y] \\ & \% // \text{Simplify} \end{aligned}$$

$$\text{Out}[81]= -y^2 (3 - 32 x y) + y^2 (1 - 16 x y) + 2 y (y - 8 x y^2)$$

$$\text{Out}[82]= 0$$

Para obtener ahora la solución de la ecuación diferencial original, empezaremos calculando

$$\text{In}[83]:= \text{expy} = \int \text{my}[x, y] \, dx$$

$$\text{Out}[83]= x y^3 - 4 x^2 y^4$$

y definimos la función

$$\begin{aligned} \text{In}[84]:= & \text{Clear}[x, y, f, h] \\ & f[x_, y_] := \text{expy} + h[y] \end{aligned}$$

$$\text{In}[86]:= f[x, y]$$

$$\text{Out}[86]= x y^3 - 4 x^2 y^4 + h[y]$$

y resolvemos finalmente la ecuación

$$\begin{aligned} \text{In}[87]:= & \text{Solve}[\partial_y f[x, y] == \text{ny}[x, y], h'[y]] \\ & h[y] = \int (h'[y] /. \%[[1]]) \, dy \end{aligned}$$

$$\text{Out}[87]= \{\{h'[y] \rightarrow 0\}\}$$

$$\text{Out}[88]= 0$$

In[89]:= f[x, y]

Out[89]=  $x y^3 - 4 x^2 y^4$

Siendo finalmente la solución de la ecuación diferencial la dada implícitamente mediante la expresión  $x y^3 - 4 x^2 y^4 = k$ , para  $k \in \mathbb{R}$  (cte).

## ■ Otras ecuaciones de primer orden (Ec. de Bernouilli)

La ecuación diferencial ordinaria  $y'(x) + P(x) y(x) = Q(x) y(x)^n$  se conoce como ecuación de Bernouilli. Aunque se puede ver con claridad que no es lineal, sí que puede reducirse a una de este tipo con el siguiente cambio de variable dependiente  $u(x) \equiv y(x)^{1-n}$ , transformándose en

$$u'(x) = (1-n) y(x)^{-n} y'(x) = (1-n) y(x)^{-n} (Q(x) y(x)^n - P(x) y(x)) = (1-n) (Q(x) - P(x) u(x))$$

Resolvamos el siguiente ejemplo:  $y'(x) + \frac{y(x)}{x} = x^3 y(x)^3$

In[90]:= Clear[x, P, Q]

P[x\_] :=  $\frac{1}{x}$   
Q[x\_] :=  $x^3$

In[93]:= n = 3;

ecdifBernouilli = y'[x] + P[x] y[x] == Q[x] y[x]^n

Out[93]=  $\frac{y[x]}{x} + y'[x] == x^3 y[x]^3$

In[94]:= ecdifmod = u'[x] == (1 - n) (Q[x] - P[x] u[x])

Out[94]=  $u'[x] == -2 \left( x^3 - \frac{u[x]}{x} \right)$

In[95]:= solEDu = DSolve[ecdifmod, u[x], x]

Out[95]=  $\{\{u[x] \rightarrow -x^4 + x^2 C[1]\}\}$


In[96]:= solEDu /. {u[x] -> y[x]^(1-n)}

Out[96]=  $\left\{ \left\{ \frac{1}{y[x]^2} \rightarrow -x^4 + x^2 C[1] \right\} \right\}$

In[97]:= y+[x\_] :=  $(-x^4 + x^2 C[1])^{-2}$   
y-[x\_] :=  $-(-x^4 + x^2 C[1])^{-2}$

In[99]:=  $\epsilon = 0.1$ ;  $c = 1$ ;

Plot[Evaluate[{y<sub>-</sub>[x], y<sub>+</sub>[x]} /. C[1] -> c], {x,  $\epsilon$ ,  $\sqrt{c}$ }]



Out[99]= - Graphics -

## ■ EDOs de segundo orden

Lo primero que hay que aclarar, lo mismo que ya hicimos en el caso de ecuaciones de primer orden no lineales, es que no existen métodos generales de resolución analítica de EDOs de segundo orden, salvo algunos casos muy concretos que iremos indicando.

### ■ Caso de coeficientes constantes

Se trata de ecuaciones diferenciales de la siguiente forma

$$y''(x) + a y'(x) + b y(x) = g(x)$$

distinguiéndose dos casos bien diferenciados, como en el caso de ecuaciones de primer orden:

- el homogéneo (cuando  $g(x) \equiv 0$ ) y
- el no homogéneo (en caso contrario).

En el caso homogéneo, supondremos soluciones de la forma  $y(x) \equiv e^{mx}$  y saquemos factor común al introducir a las derivadas en la correspondiente ecuación diferencial, obteniéndose que

$e^{mx}(m^2 + a m + b) = 0$  para todo  $x$  de la variable independiente donde se tenga definida la ec. dif.

Por ello, dado que la función exponencial no se anula, acabaremos teniendo la siguiente ecuación polinómica  $m^2 + a m + b = 0$  para el coeficiente del exponente. Esta es la denominada ecuación auxiliar, polinómica de segundo orden, y según los valores del correspondiente discriminante  $d = a^2 - 4 b$  podemos distinguir varias posibilidades:

a) discriminante positivo: en cuyo caso se tienen dos soluciones reales distintas de la ec. auxiliar,  $m = \alpha \equiv \frac{-a}{2} + \frac{\sqrt{d}}{2}$  y  $m = \beta \equiv \frac{-a}{2} - \frac{\sqrt{d}}{2}$ , siendo la solución general de nuestra ecuación diferencial no homogénea

$$y(x) = c_1 e^{\alpha x} + c_2 e^{\beta x} + \frac{1}{\alpha - \beta} (e^{\alpha x} P(x) - e^{\beta x} Q(x)), \text{ con } c_1, c_2 \in \mathbb{R} \text{ constantes.}$$

para  $P(x)$  una primitiva de la función  $e^{-\alpha x} g(x)$  y  $Q(x)$  una primitiva de  $e^{-\beta x} g(x)$ , siendo claramente ambos nulos si estuvieramos en el caso homogéneo.



b) discriminante negativo: en este caso se tienen dos soluciones imaginarias conjugadas de la ec. auxiliar,  $m = \alpha \equiv \frac{-a}{2} + i \frac{\sqrt{-d}}{2}$  y  $m = \beta \equiv \frac{-a}{2} - i \frac{\sqrt{-d}}{2}$ , siendo la solución general de nuestra ecuación diferencial no homogénea (siendo  $P(x)$  y  $Q(x)$  nulos en el caso homogéneo).

$$y(x) = e^{\frac{-a}{2}x} \left( c_1 \cos\left(\frac{\sqrt{-d}}{2}x\right) + c_2 \sin\left(\frac{\sqrt{-d}}{2}x\right) \right) + \frac{1}{\alpha - \beta} (e^{\alpha x} P(x) - e^{\beta x} Q(x)), \text{ con } c_1, c_2 \in \mathbb{R} \text{ constantes.}$$

c) discriminante nulo: teniéndose en este caso una única solución real doble de la ec. auxiliar,  $m = \alpha \equiv \frac{-a}{2}$ , siendo la solución general de nuestra ecuación diferencial no homogénea

$$y(x) = e^{\alpha x} (c_1 + c_2 x) + e^{\alpha x} P(x), \text{ con } c_1, c_2 \in \mathbb{R} \text{ constantes.}$$

Para la resolución ahora del caso homogéneo, sólo tendríamos que recordar que entonces  $P(x) = 0$ ,

## ■ Otras EDOs interesantes

### ■ Ecuación de Ricatti

Es una ecuación general del tipo  $y' + y^2 + P(x)y + Q(x) = 0$

que no puede resolverse fácilmente, salvo en casos especiales. Su importancia radica en la relación que tiene con la ecuación lineal de segundo orden  $z'' + P(x)z' + Q(x)z = 0$ , sin más que realizar el siguiente cambio  $y \equiv \frac{z'}{z}$ . Por tanto, si  $y \equiv y(x)$  es solución de la ecuación de Ricatti, entonces  $z(x) \equiv Ce^{A(x)}$  será solución de la correspondiente ecuación lineal de segundo orden, siendo  $A(x)$  una primitiva de  $y(x)$ . Así, en caso de conocerse alguna solución particular  $y_p(x)$ , entonces la solución general puede determinarse de la siguiente forma:

- escribimos  $y(x) \equiv y_p(x) + \frac{1}{v(x)}$

- derivamos ahora y sustituimos, quedando que  $v' - (P(x) + 2y_p(x))v(x) = 1$

- si conseguimos ahora  $R(x)$  una primitiva de  $P(x) + 2y_p(x)$ , bastará con resolver ahora la ecuación resultante, obteniéndose que  $v(x) = e^{R(x)}(S(x) + A(x))$ , para  $S(x)$  una primitiva de  $e^{-R(x)}$ .

### ■ Ecuación de Euler lineal

Es una ecuación lineal de orden  $n$  que se puede escribir como

$$x^n y^{(n)} + a_1 x^{n-1} y^{(n-1)} + \dots + a_{n-1} x y' + a_n y = g(x)$$

para  $a_1, \dots, a_{n-1}, a_n$  constantes reales. Y que puede reducirse fácilmente a otra de tipo lineal, pero ahora con coeficientes constantes, sin más que realizar el cambio de variable independiente  $x \equiv e^t$ , pudiéndose escribir la nueva ecuación de la siguiente forma

$$y^{(n)} + q_1 y^{(n-1)} + \dots + q_{n-1} y' + q_n y = g(e^t)$$

con  $q_1, \dots, q_{n-1}, q_n$  constantes reales también.

Veamos un ejemplo concreto, resolvamos la ecuación diferencial  $x^2 y'' + 6x y' + 6y = \frac{1}{x^2}$ .

$$\text{In}[100]:= \text{ecdifEuler} = x^2 y''[x] + 6 x y'[x] + 6 y[x] == \frac{1}{x^2}$$

$$\text{Out}[100]= 6 y[x] + 6 x y'[x] + x^2 y''[x] == \frac{1}{x^2}$$

$$\text{In}[101]:= \text{sol} = \text{DSolve}[\text{ecdifEuler}, y[x], x]$$

$$\text{Out}[101]= \left\{ \left\{ y[x] \rightarrow -\frac{1}{x^2} + \frac{C[1]}{x^3} + \frac{C[2]}{x^2} + \frac{\text{Log}[x]}{x^2} \right\} \right\}$$

$$\text{In}[102]:= \text{ecdifEulerbis} = \text{ecdifEuler} /. \{x \rightarrow e^t, y[E^t] \rightarrow w[t], E^t y'[E^t] \rightarrow w'[t], E^{2t} y''[E^t] \rightarrow w''[t]\}$$

$$\text{Out}[102]= 6 w[t] + 6 w'[t] + w''[t] == E^{-2t}$$

$$\text{In}[103]:= \text{solbis} = \text{DSolve}[\text{ecdifEulerbis}, w[t], t]$$

$$\text{Out}[103]= \left\{ \left\{ w[t] \rightarrow -\frac{E^{(-3+\sqrt{3})t}(-1+\sqrt{3})t}{2\sqrt{3}(-1+\sqrt{3})} - \frac{E^{(-3-\sqrt{3})t}(1+\sqrt{3})t}{2\sqrt{3}(1+\sqrt{3})} + E^{(-3-\sqrt{3})t} C[1] + E^{(-3+\sqrt{3})t} C[2] \right\} \right\}$$

$$\text{In}[104]:= \text{Simplify}[w[t] /. \text{solbis}[[1]]] \\ \% /. t \rightarrow \text{Log}[x]$$

$$\text{Out}[104]= -\frac{1}{2} E^{-2t} + E^{(-3+\sqrt{3})t} C[1] + E^{(-3+\sqrt{3})t} C[2]$$

$$\text{Out}[105]= -\frac{1}{2 x^2} + x^{-3-\sqrt{3}} C[1] + x^{-3+\sqrt{3}} C[2]$$

## ■ Algunas Interpretaciones geométricas

Si tenemos que resolver, o al menos estudiar, la ecuación diferencial  $y'(x) = f(x, y(x))$  y sabemos que tendrá solución única para cada condición inicial  $y(a) = \alpha$ , podríamos representar gráficamente cada una de las curvas solución, para cada valor del parámetro  $\alpha$ .


Veamos un ejemplo concreto,  $y'(x) = x - y(x)$ , cuya solución general se expresa  $y(x) = -1 + x + c_1 e^{-x}$ , como podemos comprobar fácilmente usando el comando **DSolve**.

$$\text{In}[106]:= \text{solgen} = \text{DSolve}[y'[x] == x - y[x], y[x], x]$$

$$\text{Out}[106]= \left\{ \{ y[x] \rightarrow -1 + x + E^{-x} C[1] \} \right\}$$

Mediante la orden **Evaluate** podemos realizar una representación gráfica de una tabla con las soluciones obtenidas para cada uno de los valores de la constante en un cierto rango.

```
In[107]:= grafsolgen = Plot[Table[y[x] /. solgen[[1]] /. C[1] -> c, {c, -1, 1, 0.2}] // Evaluate, {x, 0, 5},
PlotStyle -> RGBColor[1, 0, 0]]
```



```
Out[107]= - Graphics -
```

## ■ Campo de direcciones

En general, para cada par de valores  $(a, \alpha)$  podemos saber de antemano cuál será la pendiente de la única curva solución  $y_\alpha(x)$  de la ecuación diferencial que pasa por dicho punto, sin más que evaluar la función de dos variables que define dicha ecuación; es decir  $y'(a) = f(a, \alpha)$  será la correspondiente pendiente. Una técnica habitual en el estudio del comportamiento de las soluciones de una ecuación diferencial en cierta región consiste en representar en cada punto del plano un pequeño vector cuya pendiente coincida con la pendiente así obtenida. A esta representación gráfica se la denomina campo de direcciones de la ecuación diferencial.

Mathematica dispone de la orden **PlotVectorField**, incluida dentro del paquete **Graphics`PlotField`** para representar el correspondiente campo de direcciones.

```
In[108]:= ? PlotVectorField
```

PlotVectorField[f, {x, x0, x1, (xu)}, {y, y0, y1, (yu)}, (options)] produces a vector field plot of the two-dimensional vector function f.

Empecemos definiendo la función de dos variables que define la ecuación diferencial correspondiente


```
In[109]:= Clear[x, y, f]
f[x, y] := x - y
```

```
In[111]:= ? f
```

Global`f

```
f[x, y] := x - y
```


```
In[112]:= x0 = 0; x1 = 5; y0 = -2; y1 = 4;
campodir = PlotVectorField[{1, x - y}, {x, x0, x1}, {y, y0, y1}]
```



Out[112]= - Graphics -

*Podemos ahora representar conjuntamente este campo de direcciones y la familia de soluciones obtenidas anteriormente, para distintos valores de la condición inicial y vemos claramente cómo dicho campo de direcciones nos indica de una manera gráfica como se van a comportar las correspondientes soluciones de la ecuación diferencial que pasan por cada punto del plano.*

```
In[113]:= Show[grafsolgen, campodir]
```



Out[113]= - Graphics -

## ■ Un segundo ejemplo

```
In[114]:= Clear[x, y, f]
          f[x_, y_] := Sin[x + y]
```

```
In[116]:= ?f
```

```
Global`f
```

```
f[x_, y_] := Sin[x + y]
```

```
In[117]:= solgen = DSolve[y'[x] == f[x, y[x]], y[x], x]
```

```
Unique::usym : 1 is not a symbol or a valid symbol name.
```

```
Solve::verif :
```


Potential solution  $\{y[x] \rightarrow -\frac{\pi}{2} - x\}$  (possibly discarded by verifier) should be checked by hand. May require use of limits.

```
Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found.
```

```
Out[117]= {{y[x] -> \frac{\pi}{2} - x}, {y[x] -> -x - ArcCos[-\frac{2(-1+x-C[1])}{2-2x+x^2+2C[1]-2xC[1]+C[1]^2}]},
           {y[x] -> -x + ArcCos[-\frac{2(-1+x-C[1])}{2-2x+x^2+2C[1]-2xC[1]+C[1]^2}]}}
```

*Mathematica nos avisa de la posibilidad de que haya más soluciones, aparte de las obtenidas mediante el comando **DSolve**. De hecho puede probarse por otros medios que por cada punto del plano pasa una y sólo una solución de la ecuación diferencial, y vamos a representar el correspondiente campo de direcciones en un determinado rectángulo para hacernos una idea de como pueden ser dichas soluciones.*

```
In[118]:= x0 = -2 π; x1 = 2 π; y0 = -2 π; y1 = 2 π;
campodir = PlotVectorField[{1, f[x, y]}, {x, x0, x1}, {y, y0, y1}, PlotPoints -> 25]
```



```
Out[118]= - Graphics -
```

Como sabemos, el campo de direcciones de una ecuación diferencial nos puede proporcionar una valiosísima información gráfica acerca de como pueden comportarse las posibles soluciones. En este caso concreto que hemos representado, cabe pensar que haya soluciones que sean rectas (siguiendo el campo de direcciones rectilíneo que se puede entrever) y que el resto de soluciones oscilan de entre varias de estas rectas.


Esta ecuación diferencial se puede convertir en una de variables separables, sin más que efectuar el siguiente cambio de variable  $u \equiv x+y$ , con lo que  $u' \equiv 1+y' = 1 + \sin(u)$ , que podemos integrar ahora sin demasiado esfuerzo, ya que podemos escribir  $\frac{du}{\sin(u)+1} = dx$  y bastará con calcular las primitivas correspondientes

```
In[119]:= (Integrate[1/(Sin[u] + 1), u] // Simplify) == Integrate[1, x] + k
% /. {u -> x + y}
```

```
Out[119]= (1 - Cos[u] + Sin[u]) / (1 + Sin[u]) == k + x
```

```
Out[120]= (1 - Cos[x + y] + Sin[x + y]) / (1 + Sin[x + y]) == k + x
```


```
In[121]:= ecsols = Table[1 - Cos[x + y] + Sin[x + y] == (x + k) (1 + Sin[x + y]), {k, -5, 5, 0.5}];
grafsols = ImplicitPlot[ecsols, {x, x0, x1}, {y, y0, y1}, PlotStyle -> RGBColor[1, 0, 0]]
```



Out[121]= - Graphics -

*Mostramos a continuación de manera conjunta estas gráficas y las del campo de direcciones asociado, para poder apreciar más claramente la influencia de dicho campo en las soluciones de la ecuación diferencial.*

```
In[122]:= Show[campodir, grafsols]
```



```
Out[122]= - Graphics -
```


## ■ Curvas isoclinas

Para una ecuación diferencial de primer orden  $y'(x) = f(x, y(x))$ , cada curva de nivel  $f(x, y) = c$  (con  $c$ , constante) define lo que también se denomina "curva isoclina", de forma que se puede asegurar que en los puntos de una tal curva isoclina la pendiente de la solución de nuestra ecuación diferencial es siempre la misma ( $y'(x) = c$ , constante)

Vamos a considerar de nuevo la ecuación diferencial estudiada más arriba  $y'(x) = \sin(x+y(x))$  y implícitamente alguna de las curvas isoclinas asociadas.




```
In[123]:= curvasisoclinas = Table[f[x, y] == c, {c, -2, 2, 0.2}];  
grafisoclinas = ImplicitPlot[curvasisoclinas, {x, x0, x1}, {y, y0, y1}, PlotStyle -> RGBColor[0, 1, 0]]
```



Out[123]= - Graphics -

*Representemos ahora conjuntamente, tanto estas curvas isoclinas como el correspondiente campo de direcciones y las soluciones obtenidas.*

```
In[124]:= Show[campodir, grafsols, grafisoclinas]
```



```
Out[124]= - Graphics -
```

## ■ Ecuación Diferencial de una familia de curvas

*Si lo que nos planteamos ahora es el problema contrario al que hemos venido considerando hasta ahora: encontrar más bien una ecuación diferencial cuyas soluciones sean las curvas de una cierta familia  $n$ -paramétrica dada, tendremos entonces que derivar respecto de la variable independiente tantas veces como constantes haya y despejar dichas constantes, para obtener la ecuación diferencial o sistema de ecuaciones resultante.*

## ■ Ejemplo

Supongamos que necesitamos hallar la EDO cuyas soluciones son las circunferencias cuyos centros están sobre el eje  $Ox$  y son tangentes a las bisectrices de los cuadrantes primero y cuarto, cumpliéndose pues que

$$(x - x_c)^2 + y^2 = r^2 = \frac{(x_c)^2}{2}$$

ya que  $r = \text{dist}((x_c, 0), \{(x, y) \in \mathbb{R}^2 \mid y - x = 0\}) = \frac{|x_c|}{\sqrt{2}}$ .

Así pues, derivando implícitamente la ecuación anterior respecto de la variable  $x$ , obtendríamos que  $2(x - x_c) + 2y(x) y'(x) = 0$ , que podríamos simplificar como simplemente  $x + y y' = x_c$ ; quedando finalmente la siguiente ecuación diferencial, al sustituir esta expresión en la ecuación de la circunferencia del comienzo:

$$(x - (x + y y'))^2 + y^2 = \frac{((x + y y'))^2}{2} = \frac{1}{2} (x^2 + 2x y y' + (y y')^2)$$

que simplificando ahora convenientemente, obtendríamos la siguiente ecuación diferencial asociada a las circunferencias consideradas:

$$(y y')^2 + 2y^2 - 2x y y' = x^2$$

In[125]:= ? Circle

Circle[[x, y], r] is a two-dimensional graphics primitive that represents a circle of radius r centered at the point x, y.

Circle[{x, y}, {rx, ry}] yields an ellipse with semi-axes rx and ry. Circle[{x, y}, r, {theta1, theta2}] represents a circular arc.

```

In[126]:= x_c = 5;  $\theta_{\min} = \frac{\pi}{4}$ ;  $\theta_{\max} = \frac{7\pi}{4}$ ;


arco = Graphics[Circle[{x_c, 0},  $\frac{\text{Abs}[x_c]}{\sqrt{2}}$ , { $\theta_{\min}$ ,  $\theta_{\max}$ }]];

y_+[x_] := x; y_-[x_] := -x
Plot[{y_-[x], y_+[x]}, {x, -10, 10}]
Show[%, arco, AspectRatio -> Automatic]


```

General::spell1 : Possible spelling error: new symbol name "min" is similar to existing symbol "Min".

General::spell1 : Possible spelling error: new symbol name "max" is similar to existing symbol "Max".



Out[127]= - Graphics -



Out[128]= - Graphics -

## ■ Trayectorias ortogonales e isogonales

Los problemas relativos a trayectorias ortogonales no sólo son muy importantes en Matemática pura y aplicada, sino que también aparecen de forma natural en el estudio de muchos problemas de Física e Ingeniería. Por ejemplo, en la teoría de flujo de fluidos, a las familias ortogonales de curvas se las denominará líneas equipotenciales y líneas de corriente, respectivamente. En teoría del calor, éstas se denominarán líneas isotermas y líneas de flujo, etc.

Dada la ecuación de una familia de curvas uniparamétricas  $F(x,y,C) = 0$ , diremos que una curva  $\gamma$  es una trayectoria ortogonal, si ésta forma un ángulo recto con cada una de las curvas de la familia. Pero ¿cómo encontrar dichas curvas ortogonales, dada la familia de curvas? Simplemente se puede afirmar que  $F(x,y, \frac{-1}{y'}) = 0$  será la ecuación diferencial de dichas trayectorias ortogonales, ya que si  $\vec{t}$  es el vector tangente a una de las curvas de la familia y  $\vec{n}$  resultase ser el correspondiente vector normal (girado  $90^\circ$  a partir del anterior), y por tanto el vector tangente a la curva ortogonal  $\gamma$ , entonces la derivada correspondiente a esta trayectoria ortogonal se puede calcular a partir de la original, resultando ser  $\frac{-1}{y'}$ . El concepto de trayectorias isogonales viene a generalizar el concepto de curvas ortogonales, cuando se permite que el ángulo que forma dicha trayectoria con todas las curvas de la familia sea un cierto ángulo constante, pero no necesariamente de  $90^\circ$ .

Así pues, si de antemano se sabe que la ecuación diferencial satisfecha por las soluciones de nuestra ecuación diferencial original es  $y'(x) = f(x,y(x))$ , entonces bastaría con resolver la ecuación diferencial  $y'(x) = \frac{-1}{f(x,y(x))}$ , para obtener las respectivas pendientes de las trayectorias ortogonales asociadas.

## ■ Ejemplo

Para hallar las trayectorias ortogonales a la familia  $x^2 + (y - a)^2 = a^2$  empezaremos derivando esta expresión respecto a la variable independiente  $x$ , para obtener la correspondiente Ecuación Diferencial de la familia de curvas, obteniéndose

$2x + 2(y - a)y' = 0$  con lo que  $a = y + \frac{x}{y'}$  y sustituyendo esta expresión en la ecuación implícita de la familia original, nos queda  $x^2 + \left(-\frac{x}{y'}\right)^2 = \left(y + \frac{x}{y'}\right)^2$  y finalmente, simplificando adecuadamente:

$$y' = 2 \frac{xy}{x^2 - y^2}$$

y la ecuación diferencial que definirá las trayectorias ortogonales correspondientes será

$$y' = \frac{y^2 - x^2}{2xy}.$$

Intentemos ahora realizar este mismo proceso de forma mecánica, con la ayuda del cálculo simbólico del que dispone Mathematica

```
In[129]:= Clear[x, a]
          eccurvas = x^2 + (y[x] - a)^2 == a^2
```

```
Out[130]:= x^2 + (-a + y[x])^2 == a^2
```

```
In[131]:= D_x eccurvas
sol = Solve[%, a] // Simplify
```

```
Out[131]= 2 x + 2 (-a + y[x]) y'[x] == 0
```

```
Out[132]= {{a -> y[x] + \frac{x}{y'[x]}}}
```

```
In[133]:= eccurvas /. sol[[1]]
Solve[%, y'[x]]
```

```
Out[133]= x^2 + \frac{x^2}{y'[x]^2} == \left(y[x] + \frac{x}{y'[x]}\right)^2
```

```
Out[134]= {{y'[x] -> \frac{2 x y[x]}{x^2 - y[x]^2}}}
```

```
In[135]:= f[x_, y_] := \frac{2 x y}{x^2 - y^2}
```


```
In[136]:= \frac{-1}{f[x, y]}
```

```
Out[136]= -\frac{x^2 - y^2}{2 x y}
```

```
In[137]:= solcurvas = DSolve[y'[x] == f[x, y[x]], y[x], x]
```

```
Out[137]= {{y[x] -> \frac{1 - \sqrt{1 - 4 x^2 C[1]^2}}{2 C[1]}}, {y[x] -> \frac{1 + \sqrt{1 - 4 x^2 C[1]^2}}{2 C[1]}}}
```

```
In[138]:= grafcurvas =
Plot[Evaluate[
Table[
{y[x] /. solcurvas[[1]], y[x] /. solcurvas[[2]]} /. C[1] -> c, {c, 0.1, 1, 0.1}]
], {x, 0, \frac{1}{2}}, (*AspectRatio->Automatic,*)
PlotStyle -> RGBColor[0, 0, 1]]
```




```
Out[138]= - Graphics -
```

```
In[139]:= Table[{y[x] /. solcurvas[[1]], y[x] /. solcurvas[[2]]} /. C[1] -> c
, {c, 1, 2}]
```

```
Out[139]= {{1/2 (1 - sqrt(1 - 4 x^2)), 1/2 (1 + sqrt(1 - 4 x^2))}, {1/4 (1 - sqrt(1 - 16 x^2)), 1/4 (1 + sqrt(1 - 16 x^2))}}
```

```
In[140]:= grafcurvas =
Plot[Evaluate[
{y[x] /. solcurvas[[1]], y[x] /. solcurvas[[2]]} /. C[1] -> 1
],
{x, 0, 1/(2 C[1])} /. C[1] -> 1, AspectRatio -> Automatic, PlotStyle -> RGBColor[0, 0, 1]]
```



```
Out[140]= - Graphics -
```


```
In[141]:= soltrayort = DSolve[y'[x] == -1/f[x, y[x]], y[x], x]
```

```
Out[141]= {{y[x] -> -sqrt(-x^2 + x/C[1])}, {y[x] -> sqrt(-x^2 + x/C[1])}}
```

```
In[142]:= {y[x] /. soltrayort[[1]], y[x] /. soltrayort[[2]]} /. C[1] -> c
```


```
Out[142]= {-sqrt(x - x^2), sqrt(x - x^2)}
```

```
In[143]:= grafrayort =
Plot[Evaluate[
  {y[x] /. soltrayort[[1]], y[x] /. soltrayort[[2]]} /. C[1] -> 1
], {x, 0, 1}, AspectRatio -> 1]
```



Out[143]= - Graphics -

```
In[144]:= grafrayort =
Plot[Evaluate[
  Table[
    {y[x] /. soltrayort[[1]], y[x] /. soltrayort[[2]]} /. C[1] -> c, {c, 0.1, 1, 0.1}
  ], {x, 0, 1}]
```




Out[144]= - Graphics -



```

In[145]:= grafrayort =
Plot[Evaluate[
  Table[
    {y[x] /. soltrayort[[1]], y[x] /. soltrayort[[2]]} /. C[1] -> c, {c, 0.1, 1, 0.1}
  ], {x, 0, 1}, AspectRatio -> Automatic]

```





Out[145]= - Graphics -

```

In[146]:= Show[grafcurvas, grafrayort, AspectRatio -> Automatic]

```





-2

-3

Out[146]= - Graphics -

## ■ Curvas isothermas y curvas de flujo de calor

Si ahora la función  $T(x,y)$  representara la temperatura en el punto del plano  $(x,y)$ , entonces la ecuación de las curvas de nivel  $T(x,y) = c$  (para  $c \in \mathbb{R}$ , una constante), se llamarán isothermas y las correspondientes trayectorias ortogonales serán las curvas de flujo de calor. Es posible que las que conozcamos de antemano sean precisamente estas últimas y necesitemos pues calcular a partir de ellas las correspondientes isothermas. En este caso, como sabemos que ambas son ortogonales entre sí, lo que tendremos que hacer es calcular las trayectorias ortogonales de la familia de curvas que conozcamos, en este caso las de flujo de calor.

Ilustremos el proceso con un ejemplo, en el que se supone que dicha familia de curvas para el flujo de calor viene dada por la ecuación  $y^2 + 2xy - x^2 = k$  (para  $k \in \mathbb{R}$ , constante). Realizaremos todo el proceso con la ayuda de la capacidad de cálculo simbólico de Mathematica.

```
In[147]:= Clear[x, y, k]
curvasflujo = y[x]^2 + 2 x y[x] - x^2 == k
```

```
Out[148]= -x^2 + 2 x y[x] + y[x]^2 == k
```

A continuación deberemos de calcular la ecuación diferencial asociada a esta familia de curvas, para lo cuál hay que derivar y eliminar la constante que aparece.

```
In[149]:= ecdif = D[curvasflujo, x]
```

```
Out[149]= -2 x + 2 y[x] + 2 x y'[x] + 2 y[x] y'[x] == 0
```

Por último, para calcular la ecuación de las trayectorias ortogonales, que será la que nos proporcione las buscadas isothermas, lo que haremos será sustituir en esta ecuación diferencial  $y'$  por  $-\frac{1}{y'}$ , ya que sabemos que el producto de las pendientes dos curvas ortogonales del plano, debe ser -1.

```
In[150]:= ecdifisotermas = ecdif /. {y'[x] ->  $\frac{-1}{y[x]}$ } // Simplify
```

```
Out[150]=  $-\frac{2(x + y[x] + x y'[x] - y[x] y'[x])}{y'[x]} == 0$ 
```


*Resolviendo esta nueva ecuación diferencial obtendremos las curvas isotermas buscadas.*

```
In[151]:= solisotermas = DSolve[ecdifisotermas, y[x], x]
```

```
Out[151]=  $\left\{ \left\{ y[x] \rightarrow \frac{x C[1] - \sqrt{1 + 2 x^2 C[1]^2}}{C[1]} \right\}, \left\{ y[x] \rightarrow \frac{x C[1] + \sqrt{1 + 2 x^2 C[1]^2}}{C[1]} \right\} \right\}$ 
```


*Representemos gráficamente algunas de estas curvas isotermas, sin más que darle distintos valores a la constante de la que dependen.*

```
In[152]:= curvasisotermas = Table[{y[x] /. solisotermas[[1]], y[x] /. solisotermas[[2]]} /. C[1] -> c, {c, 0.1, 1, 0.1}];  
grafisotermas = Plot[Evaluate[curvasisotermas], {x, -5, 5}, PlotRange -> {-5, 5}, AspectRatio -> Automatic]
```




```
Out[152]= • Graphics •
```

```
In[153]:= curvasflujo = Table[y^2 + 2 x y - x^2 == k, {k, -10, 10, 1}];
grafsflujo = ImplicitPlot[curvasflujo, {x, -5, 5}, {y, -5, 5}, PlotStyle -> RGBColor[0, 0, 1]]
```



Out[153]= Graphics -

```
In[154]:= Show[grafisotermas, grafsflujo, AspectRatio -> Automatic]
```



Out[154]= Graphics -

## ■ Ejercicios

■ 1.- Resolver, siempre que sea posible, las siguientes EDOs de primer orden

■ a)  $2x^2 + 2y^2 + (4x y + 3y^2)y' = 0$

■ b)  $3(y - 1)^2 y' = 2 + \operatorname{sen}(x)$

■ c)  $(y^4 - y^2 + 1)y' = x^3 y - y$

■ d)  $(2 + y)y' = 2x + 3y$

■ e)  $y' = \frac{x + y}{x - y}$

■ f)  $y' = \frac{2x \cos(y) + 3x^2 y}{-x^3 + x^2 \operatorname{sen}(y) + y}$

■ g)  $(y' + y)^2 - (x - y)^2 = 0$

■ h)  $y = y' x + g(y')$  (Ecuación de Clairaut)

■ i)  $y = y' P(x) + Q(x)(y')^n$

■ 2.- Resolver, siempre que sea posible, las siguientes EDOs de segundo orden

■ a)  $y'' - 2y' - 3y = 0$

■ b)  $y'' + 4y' + 13y = 0$

■ c)  $y'' - 4y' + 4y = 0$

■ d)  $y'' - 3y' + 2y = \cos(3x)$

■ e)  $y'' + 4y' + 4y = \cos(x)$

■ f)  $y'' + y = 0$

■ **3.- Resolver el siguiente PVI**

■ a)  $y'' - y' - 2y = 4x^2$ ,  $y(0) = 0$ ,  $y'(0) = 1$ .

■ **4.- Representar el campo de direcciones asociado a las siguientes EDOs y estudiar su comportamiento y posibles soluciones a partir de este. Intentar también su resolución mediante la orden DSolve.**

■ a)  $y' = y^2 + e^{x+y}$ , para  $-5 < x < 5$ .

■ b)  $y' = \sin(x) - \frac{y}{x}$ , para  $-5 < x < 5$  (atención con los valores próximos a 0).

■ **5.- Calcular y representar las trayectorias ortogonales de las siguientes familias de curvas.**

■ a) Las parábolas  $x = k y^2$ , con  $k \in \mathbb{R}$  cte.

■ b) Las parábolas  $y = k x^2$ , con  $k \in \mathbb{R}$  cte.